SATELLITE & MESOMETEOROLOGY RESEARCH PROJECT

Department of the Geophysical Sciences
The University of Chicago

A PROGRAM FOR COMPUTER GRIDDING OF SATELLITE PHOTOGRAPHS FOR MESOSCALE RESEARCH

by

William D. Bonner



SMRP Research Paper

NUMBER 43
June 1965

MESOMETEOROLOGY PROJECT ---- RESEARCH PAPERS

1.* Report on the Chicago Tornado of March 4, 1961 - Rodger A. Brown and Tetsuya Fujita

2.* Index to the NSSP Surface Network - Tetsuya Fujita

3.* Outline of a Technique for Precise Rectification of Satellite Cloud Photographs - Tetsuya Fujita

4.* Horizontal Structure of Mountain Winds - Henry A. Brown

5.* An Investigation of Developmental Processes of the Wake Depression Through Excess Pressure Analysis of Nocturnal Showers - Joseph L. Goldman

6.* Precipitation in the 1960 Flagstaff Mesometeorological Network - Kenneth A. Styber

7.** On a Method of Single- and Dual-Image Photogrammetry of Panoramic Aerial Photographs - Tetsuya Fujita

8. A Review of Researches on Analytical Mesometeorology - Tetsuya Fujita

9. Meteorological Interpretations of Convective Nephsystems Appearing in TIROS Cloud Photographs - Tetsuya Fujita, Toshimitsu Ushijima, William A. Hass, and George T. Dellert, Jr.

10. Study of the Development of Prefrontal Squall-Systems Using NSSP Network Data - Joseph L. Goldman

11. Analysis of Selected Aircraft Data from NSSP Operation, 1962 - Tetsuya Fujita

12. Study of a Long Condensation Trail Photographed by TIROS I - Toshimitsu Ushijima

13. A Technique for Precise Analysis of Satellite Data; Volume I - Photogrammetry (Published as MSL Report No. 14) - Tetsuya Fujita

14. Investigation of a Summer Jet Stream Using TIROS and Aerological Data - Kozo Ninomiya

15. Outline of a Theory and Examples for Precise Analysis of Satellite Radiation Data - Tetsuya Fujita

* Out of print
** To be published

(Continued on back cover)

SATELLITE AND MESOMETEOROLOGY RESEARCH PROJECT
Department of the Geophysical Sciences
The University of Chicago

A PROGRAM FOR COMPUTER GRIDDING OF SATELLITE PHOTOGRAPHS FOR MESOSCALE RESEARCH

by

William D. Bonner

SMRP Research Paper #43
June 1965

The research reported in this paper has been partially supported by the National Aeronautics and Space Administration under grant NASA NsG 333 and by the National Weather Satellite Center, U. S. Weather Bureau, under grant CWB WBG - 6.

A PROGRAM FOR COMPUTER GRIDDING OF SATELLITE PHOTOGRAPHS FOR MESOSCALE RESEARCH

William D. Bonner

Department of the Geophysical Sciences

The University of Chicago

Chicago, Illinois

ABSTRACT

3/273

A computer program is described which performs the coordinate transformations used in Fujita's ~~graphical method~~ for gridding TIROS photographs. Output is a matrix of picture coordinates of latitude-longitude intersections to be plotted on a distortion-free grid. Total time required to grid a series of photographs is greatly reduced compared to the original graphical technique. Author

1. Introduction

Because TIROS is a spin-stabilized rather than an earth-oriented vehicle, the geometry for gridding of TIROS photographs is fairly complicated. Small errors in the attitude of the satellite can produce significant errors in the location of individual clouds or cloud elements on the photographs. For example, estimating the tilt to be 43° when it is actually 40° produces an error of approximately 40 n mi in the computed position of the principle point of the photograph. Even more serious errors arise from inaccuracies in picture start time of tape mode photographs on TIROS satellites - especially TIROS I, II and III. If a high degree of accuracy in the picture gridding is required, it is necessary to check the programmed picture start time through the use of landmarks on the photographs or, in the absence of landmarks, by matching satellite nadir angles as a function of time with the photogrammetrically determined tilts of individual photographs (Fujita, 1963).

Fujita (1963) describes a concise method for gridding satellite photographs which

---

The research reported in this paper has been partially supported by the National Aeronautics and Space Administration under grant NASA NsG 333 and by the National Weather Satellite Center, U. S. Weather Bureau, under grant CWB WBG - 6.

has been successfully applied in a number of meso- and synoptic-scale studies.[1] He uses a series of tilt and height grids to transform latitude and longitude intersections on an OEC (oblique equidistant cylindrical projection) chart to distortion-free grids constructed for each TIROS camera. He then transforms to fiducial grids constructed for each photograph. This method of processing TIROS photographs is apparently the most accurate developed to date (Fujita, 1964). It is, however, a time-consuming job which must be carefully and conscientiously carried out.

This report describes a computer program which duplicates Fujita's graphical procedure. It differs from previous programs[2] in that the mathematics is simplified by employing projective geometry and spherical trigonometry throughout. It was not designed to perform the entire gridding operation but only that part which can best be performed by machine. Input data is assumed to have been checked or determined by graphical techniques. The output is in terms of the distortion-free grid. Thus, hand operations are required at the beginning and end of the procedure and only the coordinate transformations are carried out by machine.

The program is written for an IBM 7094. It has been employed routinely for research studies at SMRP using the facilities of the Computation Center of the University of Chicago. Computer time for gridding all 32 pictures of a single TIROS orbit is approximately 45 sec. As with most programs, small refinements are continually being made. Although it was designed for research purposes, the program could easily be transformed to a fully operational scheme by the addition of subroutines to generate the satellite orbit and to correct for mean radial distortion.

2. Input

The following preliminary data are required:

a. Latitudes and longitudes of the subsatellite points and satellite altitudes at one-min intervals for the duration of the picture sequence plus at least one min on either side.

---

[1] Among these are SMRP Reports 9, 12, 14, 25, 28 and 35, available from SMRP, 5736 S. Woodlawn Avenue, Chicago, Illinois 60637.

[2] Dean (1961), Frankel and Bristor (1962), Mach and Gardner (1962).

b. Latitude and longitude of the spin-axis point at the initial time -- the first min of subpoint data.

c. Picture start time in min past the initial time.

Subpoint latitudes and longitudes and satellite altitudes are given in the Definitive AT Map published by NASA. Goldshlak (1962-) summarizes TIROS spin-axis data. Fujita (1963) gives spin-axis positions for TIROS I through V. The most accurate approach is to determine a fixed-earth spin-axis point directly from the photographs (Fujita, 1963).

A gridding scheme can be no better than the quality of the input data; therefore, it is important to get an accurate estimate of the spin-axis point and the picture start time by preliminary analysis on the OEC chart.

Fig. 1 shows the form used at SMRP for summarizing the input data. The information is entered on punched cards according to a format shown in the Appendix.

3. Mathematics

The program computes the tilt of each photograph from the spin-axis point and the position of the subpoint. Subsatellite points are interpolated at the time of the picture from the information at one-min intervals. Latitude of the spin-axis point is assumed constant during the time of a single sequence, however, its longitude is decreased by 0.25° per min because of the rotation of the earth. Fig. 2 shows the geometry for determining tilt τ and the azimuth of the principal line α_{PP} . From spherical triangle I in Fig. 2, we have

$$\cos\tau = \sin\phi_{SA}\sin\phi_{SP} + \cos\phi_{SA}\cos\phi_{SP}\cos(\Delta\theta) \tag{1}$$

where θ refers to longitude, ϕ to latitude and subscripts sa and sp to spin-axis and subpoints respectively.

The azimuth of the principal line, measured clockwise from true north, is given by

$$\sin(\pi-\alpha_{PP}) = \cos\phi_{SA}\frac{\sin\Delta\theta}{\sin\tau} \tag{2}$$

We determine the latitude and longitude of the principal point from the picture tilt and the azimuth of the principal line. Thus, the great circle arc δ_{PP} between subpoint and principal point is given by

$$\sin(\delta_{PP} + \tau) = \frac{H + \bar{R}}{\bar{R}} \sin\tau \qquad (3)$$

where $\bar{R}$ is the mean radius of the earth and H is the satellite height (Fujita, 1963). And from triangle II, Fig. 2, we have

$$\sin\phi_{PP} = \sin\phi_{SP} \cos\delta_{PP} + \cos\phi_{SP} \sin\delta_{PP} \cos\alpha_{PP}$$

and

$$\theta_{PP} = \theta_{SP} + \sin^{-1}\left[\frac{\sin\delta_{PP} \sin\alpha_{PP}}{\cos\phi_{PP}}\right]. \qquad (4)$$

The position of the principal point is a part of the output; it is, however, extraneous information as far as the coordinate transformations are concerned. Neglecting distortion, the parameters ϕ_{SP} , θ_{SP} , τ , α_{PP} , and H are sufficient to determine picture coordinates of arbitrary points on a spherical earth.

Coordinate transformations from latitude and longitude on the earth to Cartesian coordinates on the distortion-free grid are summarized in Figs. 3, 4, and 5. Three transformations are involved:

a. Latitude and longitude on the earth transform to nadir angle η and horizontal angle ψ measured from the principal line (Fig. 3). The equations used in this transformation are

$$\cos\delta = \sin\phi_{SP} \sin\phi_P + \cos\phi_{SP} \cos\phi_P \cos(\Delta\theta),$$

$$\sin\alpha = \frac{\cos\phi_P \sin(\Delta\theta)}{\sin\delta},$$

$$\tan\eta = \frac{\rho \sin\delta}{\rho(1 - \cos\delta) + H},$$

and

$$\psi = \alpha - \alpha_{PP}. \qquad (5)$$

b. Coordinates (η , ψ) transform to the angles (ρ , γ) on a unit sphere centered at the perspective center of the photograph (Fig. 4). We have

$$\cos\rho = \cos\eta \cos\tau + \sin\eta \sin\tau \cos\psi$$

and

$$\sin\gamma = \frac{\sin\eta \sin\psi}{\sin\rho} \qquad (6)$$

c. Coordinates (ρ , γ) transform to Cartesian coordinates on the image plane (Fig. 5). Here

$$x = f \tan\rho \sin\gamma$$

and

$$y = f \tan\rho \cos\gamma \qquad (7)$$

where f is the principal distance of the distortion-free grid.

The rather special geometry at the horizon is described in Fig. 6. As a condition for the appearance of a horizon on the photograph we have

$$(90 - \delta_H) \leq (\tau + B)$$

where δ_H is the dip (Fig. 6) and B is the camera field of view. Picture coordinates of the horizon are obtained by setting $\eta = 90 - \delta_H$ in (6) and then assuming increasing values of ψ until the computed ρ exceeds the camera field of view B .

Fig. 7 summarizes the sequence of operations carried out by the machine in solving (1) through (7).

4. Output

The machine output is in three parts:

a. A summary of the input data.

b. A series of coordinate matrices for all gridded frames within a single orbit.

c. A table giving computed position and attitude data for each frame.

A sample matrix is shown in Fig. 8. Row and column headings are latitude and longitude respectively. Matrix elements are x (upper) and y (lower) coordinates in millimeters of the indicated intersections projected onto the distortion-free grid. The principal distance f in (7) is 94.7 mm which enables the output to be plotted directly onto Fujita's (1963) grids. Zeroes in the matrix indicate that the particular intersection is outside of the camera field of view or else beyond the horizon. In order to avoid packing of grid points near the actual horizon, a fake horizon is assumed at a nadir angle that is 98 per cent of the complement of the dip angle. This is the same procedure used by Frankel and Bristor (1962) except that their percentage is 95.

If there is a visible horizon in a particular frame, horizon coordinates will be written above the longitude headings of the matrix. Since the horizon is symmetrical about the principal line, only positive values of x are computed and the complete horizon curve is obtained by plotting the points ($\pm x, y$).

The standard matrix is 18 x 18. The number of rows, however, increases in stages as the latitude of the subpoint increases in order to encompass a greater number of meridians. The grid increment is also variable: when the subsatellite latitude passes beyond 60^{o}, the longitude increment increases from 2 to 4^{o}. At 70^{o} the increment becomes 8^{o}. Thus, near-polar orbits from TIROS and Nimbus satellites can be handled with the existing output format. A latitude-longitude grid for one frame of a simulated TIROS orbit is shown in Fig. 9.

Once the matrix output has been obtained for a desired orbit, grid the pictures in the following way:

a. Construct a fiducial grid for each frame. Usually two or three fiducial grids will fit closely enough to all pictures in a single orbit.

b. If there is a horizon in a particular photograph, transfer it to the distortion-free grid. If there is no horizon, determine the principal line from a plot of the principal point track (Fujita,1963) and sketch the principal line on the distortion-free grid.

c. Plot the picture matrix on millimeter graph paper.

d. Place the translucent distortion-free grid for the particular frame on the graph paper and match the principal points and the horizon curves or principal lines.

e. Transfer latitude and longitude intersections by hand to the fiducial grid on the photograph.

Acknowledgements:

I wish to thank Mrs. G. Baralt for her assistance in "de-bugging" various phases of the program and Mr. James Arnold for useful discussions concerning the form of the output. Thanks are due, of course, to Prof. Fujita upon whose work the entire gridding procedure is based.

REFERENCES

Dean, C., 1961: Grid program for TIROS II pictures. Allied Research Ass., Inc. Final Report, U. S. Weather Bureau, Contract No. CWB 10023.

Fujita, T., 1963: A technique for precise analysis of satellite data; Vol. I - Photogrammetry. MSL Report 14, U. S. Weather Bureau, 106 pp.

_____, 1964: Evaluation of errors in the graphical rectification of satellite photographs. SMRP Research Paper 30, Univ. of Chicago, 23 pp.

Frankel, M., and C. L. Bristor, 1962: Perspective locator grids for TIROS pictures. MSL Report 11, U. S. Weather Bureau, 38 pp.

Goldshlak, L., 1962 - : TIROS Attitude Summaries. ARACON Geophysics Company, Concord, Mass.

Mach, R. E., and T. L. Gardner, 1962: Rectification of satellite photography by digital techniques. IBM Journal of Research and Development, Vol. 6, No. 3.

APPENDIX

THE FORTRAN PROGRAM

The program was written in FORTRAN II for the IBM 7090 or 7094. With minor changes it should be adaptable to smaller machines. Inverse trigonometric functions are an essential part of the program, however, and these are not always included as library subroutines in the smaller systems.

At SMRP radiation data are typically analyzed in conjunction with the photographs and a separate program has been written which locates geographically the analog radiation readings. Radiation and photo-gridding routines are actually subroutines called for by a short main program. However, the photo-gridding program given here is complete in itself. The program is separated into logical units. A short summary of the purpose of the unit precedes the program statements and definitions of the FORTRAN names for variables, constants and subroutines follow.

A. Read the identifier and subpoint cards. Summarize the input data.

```
      F=94.7
      C1=.0174532925
      C2=57.2957795
      C3=6367.65
      PI=3.1415027536
      DIMENSION LAT(20),LONG(30),CDIF2(30),SDIF2(30),CLAT(30),SLAT(20),
     1 NXHOR(10),NYHOR(10),NX(20,30),NY(20,30),TABLE(9,50),SPDATA(3,50)
  100 READ INPUT TAPE 5,1,JTIROS,JORBIT,JM,JD,JY,SA1,REFSA2,REFT,NO,
     1 NFORK
    1 FORMAT (I2,I5,3I2,2F5.1,F5.0,2I2)
  200 READ INPUT TAPE 5,2,((SPDATA(M,N),M=1,3),N=1,NO)
    2 FORMAT (2F5.1,F5.0)
  300 WRITE OUTPUT TAPE 6,3,JTIROS,JORBIT,JM,JD,JY,SA1,REFSA2,REFT,
     1 ((SPDATA(M,N),M=1,3),N=1,NO)
    3 FORMAT (1H1,15X,31HSUMMARY OF INPUT DATA TIROS NO.,I2,16H ORBIT N
     1 O.(A/O),I5,5X,5HMONTH,I3,5X,5H DAY,I3,9H YEAR 19,I2//28X,18HSPIN
     2  AXIS LATITUDE,F5.1,3HDEG,24H   SPIN AXIS  LONGITUDE,F6.1,3H  AT,
     3 F6.0,1HZ//43X,35HORBITAL DATA FROM DEFINITIVE AT MAP/20X,80HONE-MI
     4 NUTE TIME INTERVALS STARTING FROM THE TIME OF THE SPIN AXIS DETERM
     5 INATION.//48X,24H        LAT.        LONG.          HT./(48X,2F8.1,F8.0))
```

Definitions

F	----------	Principal distance of distortion-free grid or photo.
C1	----------	Conversion factor degrees to radians.
C2	----------	Conversion factor radians to degrees.
C3	----------	Mean radius of the earth.
JTIROS	----------	TIROS Number.
JORBIT	----------	Orbit Number (Actual Orbit).
JM,JD,JY	----------	Month, Day, Year.
SA1	----------	Latitude of spin-axis point.
REF SA2,REFT	----------	Long. of spin-axis point at initial time, REFT.
NO	----------	Number of minutes of subpoint data to follow.

SPDATA (1,N)	----------	Latitude of subpoint at time REFT plus (N-1) minutes.
SPDATA (2,N)	----------	Longitude of subpoint.
SPDATA (3,N)	----------	Altitude of satellite in km.

B. Read the "picture" card. Write identifying information and explanation of format.

```
      READ INPUT TAPE 5,1,IMAGE, START,TDIF,NUMBER,MODE
    1 FORMAT  (I2,F5.2,F3.2,2I2)
      ISTEP=2
      DVIEW=50.0
      SINSA1=SINF(C1*SA1)
      COSSA1=SQRTF(1.-SINSA1**2)
      VIEW=C1*DVIEW
  500 WRITE OUTPUT TAPE 6,5,JTIROS,JORBIT,JM,JD,JY,ISTEP,F
    5 FORMAT  (1H1,20X,5HTIROS,I3,55X,12HORBIT NUMBER,I5/47X,5HMONTH,I2,3
     1 X,3HDAY,I2,3X,7HYEAR 19,I2///32X,20HIMAGE COORDINATES OF,I2, 41H DE
     2 G LATITUDE AND LONGITUDE INTERSECTIONS//40X,37HCARTESIAN COORDINAT
     3 ES CENTERED AT THE/40X,40HPRIMARY POINT,Y AXIS ALONG PRIMARY LINE,
     4 /40X,41HX(UPPER) AND Y(LOWER) IN MM TO BE PLOTTED/40X,40HON DISTOR
     5 TION-FREE GRID OR PHOTO WITH F=,F5.1,3HMM.)
```

Definitions

IMAGE	----------	Number of the first frame. Will typically be 32 for tape mode, 1 for direct.
START	----------	Picture start time in minutes (tenths and hundredths) past the initial time.
TDIF	----------	Time interval in minutes between successive frames.
NUMBER	----------	Number of photos to be gridded.
MODE	----------	Plus 1 for direct, -1 for tape.
ISTEP	----------	Latitude increment for photo grids.
DVIEW	----------	Field of view of camera in degrees.

C. Set up a two-dimensional array TABLE giving frame number, time (with respect to initial time), subpoint latitude and longitude and satellite altitude for each picture to be gridded. Latitude, longitude and altitude are obtained by linear interpolation in matrix SPDATA.

```
      DO 6 J=1,NUMBER
      TABLE(1,J)=IMAGE+(J-1)*MODE
      DUMMY=J-1
      TABLE(2,J)=START+(DUMMY*TDIF)
      DUMMY=INTF(TABLE(2,J))
      R=TABLE(2,J)-DUMMY
      INDEX=DUMMY+1.
      TABLE(3,J)=SPDATA(1,INDEX)+R*(SPDATA(1,INDEX+1)-SPDATA(1,INDEX))
      IF (ABSF (SPDATA(2,INDEX+1)-SPDATA(2,INDEX))-100.)4,2,2
    2 D1=SPDATA(2,INDEX)
      D2=SPDATA(2,INDEX+1)
      IF(D1-D2)210,210,212
  210 D2=360.
      TABLE(4,J)=D1+R*(D2-D1)
      IF(TABLE(4,J))220,6,6
  220 TABLE(4,J)=TABLE(4,J)+360.
      GO TO 6
  212 D2=D2+360.
      TABLE(4,J)=D1+R*(D2-D1)
      IF(TABLE(4,J)-360.)6,222,222
  222 TABLE(4,J)=TABLE(4,J)-360.
      GO TO 6
    4 TABLE(4,J)=SPDATA(2,INDEX)+R*(SPDATA(2,INDEX+1)-SPDATA(2,INDEX))
    6 TABLE(5,J)=SPDATA(3,INDEX)+R*(SPDATA(3,INDEX+1)-SPDATA(3,INDEX))
```

Definitions

TABLE (1,J)	----------	Frame number of jth frame.
TABLE (2,J)	----------	Picture time in minutes past initial time.
TABLE (3,J)	----------	Subpoint latitude at picture time.
TABLE (4,J)	----------	Subpoint longitude at picture time.
TABLE (5,J)	----------	Satellite altitude in km at picture time.

D. Keep a count of the number of pictures that have been gridded. Define the frame number, time, subpoint latitude and longitude and satellite altitude for the particular frame to be gridded. Do the preliminary calculations for computing tilt and the azimuth of the primary line.

```
200 JCOUNT=JCOUNT+1
    IF(JCOUNT-NUMBER)8,8,800
  8 IMAGE=TABLE(1,JCOUNT)
    DELTAT=TABLE(2,JCOUNT)
    SP1=TABLE(3,JCOUNT)
    SP2=TABLE(4,JCOUNT)
    H=TABLE(5,JCOUNT)
    JSTEP=2
 10 SA2=REFSA2-.25*DELTAT
    IF(SA2)11,17,17
 11 SA2=SA2+360.
 17 DUMMY=C1*SP1
    COSSP1=COSF(DUMMY)
    SINSP1=SINF(DUMMY)
    DIF2=C1*ABSF(SA2-SP2)
    FSP2=SP2
    FSA2=SA2
    IF(DIF2-PI)16,12,12
 12 IF(SA2-SP2)13,14,14
 13 FSP2=SP2-360.
    GO TO 15
 14 FSA2=SA2-360.
 15 DIF2=2.*PI-DIF2
```

Definitions

JCOUNT	----------	Position in array TABLE of data referring to particular frame being gridded.
IMAGE	----------	Number of particular frame being gridded.
DELTAT	----------	Picture time in minutes past REFT.
DIF2	----------	Difference in radians between subpoint and spin-axis longitudes.
H	----------	Satellite height in km.
SP1	----------	Subpoint latitude.
SP2	----------	Subpoint longitude.

SA2	----------	Spin-axis longitude and picture time.
JSTEP	----------	Longitude increment of photo grid.

E. Compute the picture tilt and the azimuth of the principal line.

```
16  CTILT=(SINSP1*SINSA1)+(COSSA1*COSSP1*COSF(DIF2))
    IF(ABSF(CTILT)-1.)20,18,18
18  CTILT=1.
    STILT=0.
    TILT=0.
    AZPL=0.
    GO TO 30
20  STILT=SQRTF(1.-CTILT**2)
    TILT=ACOSF(CTILT)
    ARG=COSSA1*SINF(DIF2)/STILT
    IF(ARG-1.)22,21,21
21  AZPL=PI/2.
    GO TO 30
22  AZPL=ASINF(ARG)
    IF(SINSP1*CTILT-SINSA1)24,21,26
24  AZPL=PI-AZPL
26  IF(FSA2-FSP2)30,30,27
27  AZPL=2.*PI-AZPL
```

Definitions

TILT	----------	Picture tilt, τ .
AZPL	----------	Azimuth of the principal line, α_{pp} .

F. Determine whether or not the principal point of the photo will be on the earth. If not, set its latitude and longitude equal to minus zero. If it is on the earth, compute its latitude and longitude.

```
30 RATIO=C3/(C3+H)
   COMDIP=ASINF(RATIO)
   CAZPL=COSF(AZPL)
   SAZPL=SINF(AZPL)
   IF(COMDIP-TILT) 32,32,31
32 PM1=-0.
   PM2=-0.
   GO TO 33
31 D=ASINF(STILT/RATIO)-TILT
   SIND=SINF(D)
   SINPM1=(SINSP1*COSF(D))+(COSSP1*SIND*CAZPL)
   COSPM1=SQRTF(1.-SINPM1**2)
   PM1=C2*ASINF(SINPM1)
   DIF2=C2*ASINF(SIND*SINF(AZPL)/COSPM1)
   THETA=SP2+DIF2
   PM2=DCHK(THETA)
```

Definitions

COMDIP	----------	Complement of the dip, δ_H .
PM1	----------	Latitude of the principal point.
PM2	----------	Longitude of the principal point.
D	----------	Great circle distance subpoint to principal point.
DIF2	----------	Difference in degrees between subpoint and principal point longitudes.
DCHK(THETA)	----------	Subroutine which insures that $0 \le$ THETA ≤ 360.

G. Store the latitude and longitude of the principal point, the azimuth of the principal line and the picture tilt in their proper positions in array TABLE for later output.

```
33 TABLE(6,JCOUNT)=PM1
   TABLE(7,JCOUNT)=PM2
   DAZPL=C2*AZPL
   DTILT=C2*TILT
   TABLE(8,JCOUNT)=DTILT
   TABLE(9,JCOUNT)=DAZPL
```

H. Determine whether or not there will be a horizon on the photo. If not, print out the heading for the picture grid, the orientation information and the words no visible horizon. If there is a visible horizon, compute (x,y) coordinates of the horizon curve at 10^o increments of horizontal angle, ψ . Round to the nearest integer and write out the horizon coordinates along with the picture heading and the orientation data.

```
   XANGLE=VIEW+TILT
   YANGLE=VIEW-TILT
   IF(COMDIP-XANGLE)35,35,50
35 A=COSF(COMDIP)
   B=SQRTF(1.-A**2)
   AZIM=0.
   DO 38 I=1,6
   COSRHO=(A*CTILT)+(B*STILT*COSF(AZIM))
   SINRHO=SQRTF(1.-COSRHO**2)
   SALFA=B*SINF(AZIM)/SINRHO
   CONST=1.
   IF(COSRHO*CTILT-A)36,37,37
36 CONST=-1.
37 DUMMY=F*SINRHO/COSRHO
   FIX=DUMMY*SALFA
   NXHOR(I)=I ROUND(FIX)
   FIX=CONST*DUMMY*SQRTF(1.-SALFA**2)
   NYHOR(I)=I ROUND(FIX)
38 AZIM=AZIM+10.*C1
40 WRITE OUTPUT TAPE 6,41,IMAGE,SP1,SP2,PM1,PM2,DTILT,DAZPL,
  1 (NXHOR(I),I=1,6),(NYHOR(I), I=1,6)
41 FORMAT  (1H1,10X,11HPICTURE NO.,I3,5X,3HSP1,F6.1,5X,3HSP2,F6.1,5X,3
  1 HPM1,F6.1,5X,3HPM2,F6.1,5X,4HTILT,F5.1,5X,4HAZPL,F6.1/50X,19HHORIZ
  2 ON COORDINATES/(30X,6I10))
   D2=PI/2.-COMDIP
   GO TO 60
50 WRITE OUTPUT TAPE 6,51,IMAGE,SP1,SP2,PM1,PM2,DTILT,DAZPL
51 FORMAT  (1H1,10X,11HPICTURE NO.,I3,5X,3HSP1,F6.1,5X,3HSP2,F6.1,5X,3
  1 HPM1,F6.1,4X,3HPM2,F6.1,5X,4HTILT,F5.1,5X,4HAZPL,F6.1/50X,18HNO VI
  2 SIBLE HORIZON//)
```

Definitions

AZIM	----------	Horizontal angle ψ measured in radians from the principal line. See Fig. 4 in text.
COSRHO	----------	cos (ρ) where ρ is the radial angle on a unit sphere centered at the perspective center of the photograph.
SALFA	----------	sin (γ) where γ is the azimuth on the unit sphere. See Fig. 4 in text.
IROUND(FIX)	----------	Subroutine which rounds to the nearest integer and converts to fixed point for output.
NXHOR	----------	x coordinate of horizon point.
NYHOR	----------	y coordinate of horizon point.

I. Compute the mid-point for the photo-grid array. This will be a point on the primary line roughly mid-way between the horizon (if one exists) and the bottom of the photo.

```
   D2=ASINF(SINF(XANGLE)/RATIO)-XANGLE
60 D1=ASINF(SINF(YANGLE)/RATIO)-YANGLE
   D3=(D2-D1)/2.
   SIND=SINF(D3)
   SINP1=(SINSP1*COSF(D3))+(COSSP1*SIND*CAZPL)
   COSP1=SQRTF(1.-SINP1**2)
   P1=C2*ASINF(SINP1)
   DIF2=C2*ASINF(SIND*SAZPL/COSP1)
   THETA=SP2+DIF2
   P2=DCHK(THETA)
```

Definitions

D1	----------	Great circle arc d_1 in Fig. 6 in text.
D2	----------	Great circle arc d_2 in Fig. 6 in text.
D3	----------	Great circle arc d_3 in Fig. 6 in text.

P1	----------	Latitude of "mid-point."
P2	----------	Longitude of "mid-point."

J. Select the proper grid dimension and longitude increment on the basis of the subpoint latitude.

```
     DUMMY=ABSF(SP1)
     IF(DUMMY-40.)68,64,64
 64  JP=INTF(DUMMY/10.)-3.
     GO TO (641,642,643,644,644),JP
641  MID=13
     JNO=11
     GO TO 69
642  MID=15
     JNO=13
     GO TO 69
643  MID=15
     JNO=13
     JSTEP=4
     GO TO 69
644  MID=15
     JNO=13
     JSTEP=8
     GO TO 69
 68  MID=10
     JNO=8
     JP=6
```

Definitions

MID	----------	Column in the longitude, latitude matrix which will serve as the location for the starting longitude.
JNO	----------	Number of longitude columns beyond MID that the grid will extend.
JSTEP	----------	Longitude increment.

K. Set the central values of latitude and longitude for the photo-grid matrix equal to the even integers nearest to the latitude and longitude of the previously computed mid-point (P1, P2). Compute the sine and cosine functions for the central latitude and longitude required for the coordinate transformations to follow.

```
 69  P2=INTF(P2)
     DUMMY=P2/2.
     IF(INTF(DUMMY)-DUMMY)691,692,691
691  LONG(MID)=P2+1.
     GO TO 693
692  LONG(MID)=P2
693  P1=INTF(P1)
     DUMMY=P1/2.
     IF(INTF(DUMMY)-DUMMY)694,695,694
694  LAT(10)=P1+1.
     GO TO 70
695  LAT(10)=P1
 70  DUMMY=LAT(10)
     DUMMY=C1*DUMMY
     SLAT(10)=SINF(DUMMY)
     CLAT(10)=COSF(DUMMY)
     DUMMY=LONG(MID)
     DIF2=ABSF(DUMMY-SP2)
     CDIF2(MID)=COSF(C1*DIF2)
     SDIF2(MID)=SQRTF(1.-CDIF2(MID)**2)
```

Definitions

LONG	----------	One-dimensional longitude array.
LAT	----------	One-dimensional latitude array.
SLAT	----------	Sine latitude array.
CLAT	----------	Cosine latitude array.
CDIF2	----------	Cosine of the longitude difference between subpoint and grid point. DIF2 corresponds to $\Delta\theta$ in Fig. 3 in text.
SDIF2	----------	Sine of the longitude difference between subpoint and grid point.

L. Set up the latitude, cosine latitude, and sine latitude arrays.

```
      DO  72 J=11,18
      LAT(J)=LAT(J -1)-ISTEP
      DUMMY=LAT(J)
      DUMMY=C1*DUMMY
      CLAT(J)=COSF(DUMMY)
  72  SLAT(J)=SINF(DUMMY)
      DO  74  J=1,9
      M=10-J
      LAT(M)=LAT(M+1)+ISTEP
      DUMMY=LAT(M)
      DUMMY=C1*DUMMY
      CLAT(M)=COSF(DUMMY)
  74  SLAT(M)=SINF(DUMMY)
```

M. Set up the longitude, CDIF2 and SDIF2 arrays.

```
      JBOT=MID+1
      JTOP=MID+JNO
      DO 76 J=JBOT,JTOP
      LONG(J)=LONG(J -1)+JSTEP
      DUMMY=LONG(J)
      DIF2=C1*ABSF(DUMMY-SP2)
      IF(LONG(J)-360)75,73,73
  73  LONG(J)=LONG(J)-360
  75  CDIF2(J)=COSF(DIF2)
  76  SDIF2(J)=SQRTF(1.-CDIF2(J)**2)
      JTOP=MID-1
      DO 80 J=1,JTOP
      M=MID-J
      LONG(M)=LONG(M+1)-JSTEP
      DUMMY=LONG(M)
      DIF2=C1*ABSF(DUMMY-SP2)
      IF(LONG(M))78,79,79
  78  LONG(M)=LONG(M)+360
  79  CDIF2(M)=COSF(DIF2)
  80  SDIF2(M)=SQRTF(1.-CDIF2(M)**2)
```

N. Compute the (x,y) coordinates of the matrix of latitude, longitude intersections. Starting at LAT(10) and LONG(MID), compute (η, ψ) then (ρ, γ) then (x,y) coordinates on the image plane. Keeping the latitude constant, but increasing the longitude, continue the calculations until the horizon is reached or until ρ exceeds the field of view of the camera. When either of these occur, return to the starting longitude and repeat the calculation for decreasing increments of longitude, again, until the horizon or the edge of the field of view has been reached. Repeat the above procedure at each of the latitude lines until the central longitude is off the earth or off the picture.

```
  90  ISTART=10
      KFORK=1
  92  ISTOP=ISTART+8
      DO 162 I=ISTART,ISTOP
      GO TO (94,96),KFORK
  94  K=I
      GO TO 97
  96  K=10-I
  97  JSTART=MID
      MFORK=1
  98  JSTOP=JSTART+JNO
      DO 151 J=JSTART,JSTOP
      GO TO (110,112),MFORK
 110  M=J
      GO TO 114
 112  M=MID-J
 114  CDIST=SLAT(K)*SINSP1+CLAT(K)*COSSP1*CDIF2(M)
      IF(CDIST-1.)116,116,115
 115  CDIST=1.
 116  SDIST=SQRTF(1.-CDIST**2)
      XNADIR=ATANF(C3*SDIST/(C3*(1.-CDIST)+H))
      IF(XNADIR-0.98*COMDIP)117,117,160
 117  CNADIR=COSF(XNADIR)
      SNADIR=SINF(XNADIR)
      SGAMMA=CLAT(K)*SDIF2(M)/SDIST
      IF(ABSF(SGAMMA)-1.)119,119,113
 113  SGAMMA=1.
 119  GAMMA=ASINF(SGAMMA)
      IF(CDIST*SINSP1-SLAT(K))120,120,118
 118  GAMMA=PI-GAMMA
 120  DUMMY=LONG(M)
      IF(SP2-DUMMY)124,124,122
 122  GAMMA=2.*PI-GAMMA
 124  PSI=GAMMA-AZPL
 130  COSRHO=CNADIR*CTILT+SNADIR*STILT*COSF(PSI)
1301  RHO=ACOSF(COSRHO)
      IF(RHO-VIEW)132,132,160
 132  SALFA=SINF(PSI)*SNADIR/SQRTF(1.-COSRHO**2)
      IF(ABSF(SALFA)-1.)136,136,134
 134  SALFA=INTF(SALFA)
 136  ALFA=ASINF(SALFA)
      IF(COSRHO*CTILT-CNADIR)140,150,150
 140  IF(ALFA)142,144,144
 142  ALFA=(-1.)*(PI+ALFA)
      GO TO 150
```

```
144   ALFA=PI-ALFA
150   DUMMY=F*TANF(RHO)
      FIX=DUMMY*SALFA
      NX(K,M)=IROUND(FIX)
      FIX=DUMMY*COSF(ALFA)
151   NY(K,M)=IROUND(FIX)
152   IF(JSTART-MID)162,154,162
154   JSTART=1
      MFORK=2
      GO TO 98
160   IF(J-MID)152,164,152
162   CONTINUE
164   IF(ISTART-10)300,165,300
165   ISTART=1
      KFORK=2
      GO TO 92
```

Definitions

CDIST	----------	Cosine of the great circle distance δ between subpoint and grid point. See Fig. 3 in text.
XNADIR	----------	Nadir angle η of grid point viewed from satellite. See Fig. 3.
GAMMA	----------	Horizontal angle of grid point measured from true North. Equivalent to the azimuth α in Fig. 3.
PSI	----------	Horizontal angle ψ of grid point. See Fig. 3.
RHO	----------	Radial angle, ρ . See Fig. 4 in text.
ALFA	----------	Azimuth γ of the projection of the grid point onto the unit sphere. See Fig. 4.
NX	----------	x coordinate of grid point.
NY	----------	y coordinate of grid point.

O. Select the output statement with the proper dimensions and the correct longitude increment and print out the picture matrix.

```
300  GO TO (310,330,330,330,330,340),JP
310  WRITE OUTPUT TAPE 6,311,(LONG(M),M=1,24),(LAT(I),(NX(I,M),M=1,24),
    1(NY(I,M),M=1,24),I=1,18)
311  FORMAT (14X,24I4,18(//10X,25I4/14X,24I4))
     GO TO 400
330  WRITE OUTPUT TAPE 6,331,(LONG(M),M=1,28),(LAT(I),(NX(I,M),M=1,28),
    1(NY(I,M),M=1,28),I=1,18)
331  FORMAT  (6X,28I4,18(//2X,29I4/6X,28I4)
     GO TO 400
340  WRITE OUTPUT TAPE 6,341(LONG(M),M=1,18),(LAT(I),(NX(I,M),M=1,18),
    1(NY(I,M),M=1,18),I=1,18)
341  FORMAT  (26X,18I4,18(//22X,19I4/26X,18I4))
```

P. Re-set all matrix elements equal to 0. Go back to compute the photo-grid for the next frame stored in TABLE. If there are no more frames to be gridded on this orbit, convert all times in TABLE to the proper form for output and print out the entire table as a summary of the photo data. Read new data card. Exit if no more data.

```
400  DO 175 I=1,18
     DO 175 J=1,30
     NX(I,J)=0
175  NY(I,J)=0
     GO TO 200
800  DO 180 J=1,NUMBER
     ZTIME=REFT+TABLE(2,J)
180  TABLE (2,J)=CONVRT(ZTIME)
     WRITE OUTPUT TAPE6,801,JORBIT,JTIROS,JM,JD,JY,((TABLE(I,J),I=1,9)
    1,J=1,NUMBER)
801  FORMAT (1H1,15X,38HSUMMARY OF PHOTO-GRID OUTPUT ORBIT NO.,I5,14H
    1      TIROS NO., I2,5X,5HMONTH,I3,5H   DAY,I3,9H   YEAR   19,I2//18X,
    285HFRAME        TIME          SP1          SP2       HEIGHT          PM1          P
    3M2           TILT            AZPL//(20X,F3.0,F10.2,2F10.1,F10.0,4F10.1))
     GO TO 100
     END(1,0,0,0,0,0,1,0,0,1,0,0,0,0,0)
```

The program just described assumes the existence of certain built-in and library functions. In addition, three FORTRAN functions were defined. Those functions normally a part of the FORTRAN system are as follows:

ABSF	----------	Absolute value function.
INTF	----------	Integer function. Converts to floating point integer by simply truncating the decimal portion.
SIGNF	----------	Transfer of sign. Requires two arguments , it assigns to the first, the sign of the second.

The following library functions were used:

COSF	----------	Cosine of an argument in radians.
SINF	----------	Sine of an argument in radians.
TANF	----------	Tangent of an argument in radians.
ACOSF	----------	Arccosine.
ATANF	----------	Arctangent.
ASINF	----------	Arcsine.
SQRTF	----------	Square root of a positive number.
COTANF	----------	Cotangent.

The three handwritten FORTRAN functions are described below:

DCHK(THETA). Checks to make certain that an argument THETA lies between 0 and 360^o.

FUNCTION DCHK(THETA)

```
  FUNCTION DCHK(THETA)
  IF(THETA-360.)4,2,2
2 DCHK=THETA-360.
  RETURN
4 IF(THETA)6,8,8
6 DCHK=THETA+360.
  RETURN
8 DCHK=THETA
  RETURN
  END(1,0,0,0,0,0,1,0,0,1,0,0,0,0,0,)
```

IROUND(FIX). Takes a floating point argument FIX and rounds it off to the nearest fixed point integer.

```
      FUNCTION   IROUND(FIX)
      AFIX=INTF(FIX)
      IF(ABSF(FIX-AFIX)-.5)10,5,5
    5 BFIX=ABSF(AFIX)+1.
      IROUND=SIGNF(BFIX,AFIX)
      RETURN
   10 IROUND=AFIX
      RETURN
      END(1,0,0,0,0,0,1,0,0,0,0,0,)
```

CONVRT(ZTIME). Takes a time ZTIME in hours and minutes and makes certain that the minutes do not exceed 59 and the hours 23.

```
      FUNCTION CONVRT(ZTIME)
      XTIME=ZTIME/100.
      YTIME=INTF(XTIME)
      DUMMY=XTIME-YTIME
      IF(DUMMY-.60)4,2,2
    2 DUMMY=DUMMY-.60
      XTIME=YTIME+1.+DUMMY
    4 IF(XTIME-24.)8,6,6
    6 XTIME=XTIME-24.
    8 CONVRT=100.*XTIME
      RETURN
      END(1,0,0,0,0,0,1,0,0,1,0,0,0,0,0)
```

TIROS No. ______
Month ______ Day ______ Year 196___
Actual Orbit No. A/O ______

TSA at the Initial Time

Initial Time (t_I)	Z
Latitude (—for south)	.
Longitude (east)	.

Orbital Data from Definitive AT Map

t	ϕ^{TSP}	θ^{TSP}	H^{SAT}
t_I →	.	.	

SCAN POINT CALCULATION ☐
PHOTO GRID CALCULATION ☐
Readout Orbit No. R/O ______
Picture Start Time(t_s) ☐☐ hr ☐☐.☐☐ min
Exposure time of first direct or last tape picture (always positive)

$$\Delta t_e = t_e - t_I$$

t_e is the exposure time of each frame

CAMERA No.

Frame No. Tape	Frame No. Direct	Δt_e Positive
32	1	.
31	2	.
30	3	.
29	4	.
28	5	.
27	6	.
26	7	.
25	8	.
24	9	.
23	10	.
22	11	.
21	12	.
20	13	.
19	14	.
18	15	.
17	16	.
16	17	.
15	18	.
14	19	.
13	20	.
12	21	.
11	22	.
10	23	.
9	24	.
8	25	.
7	26	.
6	27	.
5	28	.
4	29	.
3	30	.
2	31	.
1	32	.

specify with a circle

1/10 min
1/100 min

TIROS ______ A/O ______ R/O ______ Month ______ Day ______ Year 196___
Calculation Requested by ______
Date of Request ______
For Research on ______

Make 3 copies, send original to Fujita, 1st copy to computation section, keep 2nd and 3rd copies to be filed with machine printout.

DO NOT FILL IN

Fig. 1. Form used at SMRP for listing the input data prior to punching.



Fig. 2. Geometry for determining tilt, azimuth of the principal line, and location of the principal point.



Fig. 3. Coordinate transformation at an arbitrary point P from latitude and longitude to nadir angle η and horizontal angle ψ . Triangle is solved first to give great circle distance δ and azimuth α .



Fig. 4. Geometry for transformation from (η , ψ) to coordinates (ρ , γ) on a unit sphere centered at the perspective center of the photograph.



Fig. 5. Transformation from (ρ , γ) to (x, y) co-ordinates on the distortion-free image plane.



Fig. 6. Horizon geometry and midpoint calculation (see section I in Appendix).



Fig. 7. Flow chart.

PICTURE NO. 1 SP1 6.8 SP2 6.7 PM1 5.9 PM2 8.0 TILT 12.7 AZPL 125.5

HORIZON COORDINATES

			0			23		46			68		90			112		
			113			111		107			100		89			75		
	358	0	2	4	6	8	10	12	14	16	18	20	22	24	26	28	30	32
18	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
16	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
14	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
12	0	0	0	-41	-56	-69	-80	-88	-96	-101	-105	0	0	0	0	0	0	0
	0	0	0	-102	-74	-49	-26	-6	11	26	38	0	0	0	0	0	0	0
10	0	0	1	-16	-32	-46	-58	-68	-76	-83	-89	-93	0	0	0	0	0	0
	0	0	-111	-83	-56	-32	-11	8	24	38	49	59	0	0	0	0	0	0
8	0	0	26	8	-8	-23	-36	-47	-57	-65	-72	-77	-81	0	0	0	0	0
	0	0	-90	-64	-39	-16	4	22	37	49	60	68	75	0	0	0	0	0
6	0	0	49	31	14	-1	-15	-28	-38	-47	-55	-61	-67	-71	0	0	0	0
	0	0	-71	-46	-22	-1	18	34	48	60	69	77	83	87	0	0	0	0
4	0	0	69	51	34	19	4	-9	-20	-30	-39	-46	-52	-57	0	0	0	0
	0	0	-52	-29	-7	13	30	45	58	69	77	84	89	93	0	0	0	0
2	0	0	87	69	53	37	22	8	-4	-14	-24	-32	-38	-44	-49	0	0	0
	0	0	-35	-13	7	25	41	55	67	76	84	90	95	98	101	0	0	0
0	0	0	102	85	68	53	38	24	12	1	-9	-18	-25	-32	0	0	0	0
	0	0	-20	0	19	36	51	63	74	83	90	96	100	103	0	0	0	0
-2	0	0	0	98	82	67	52	38	26	14	4	-5	-13	-20	0	0	0	0
	0	0	0	12	29	45	58	70	80	88	94	100	103	106	0	0	0	0
-4	0	0	0	108	93	78	64	51	38	27	16	7	-1	0	0	0	0	0
	0	0	0	23	38	52	65	75	84	92	98	103	106	0	0	0	0	0
-6	0	0	0	0	102	88	74	61	49	37	27	18	0	0	0	0	0	0
	0	0	0	0	45	58	70	80	88	95	100	105	0	0	0	0	0	0
-8	0	0	0	0	0	0	83	70	58	47	37	27	0	0	0	0	0	0
	0	0	0	0	0	0	74	83	90	97	102	106	0	0	0	0	0	0
-10	0	0	0	0	0	0	0	0	0	55	0	0	0	0	0	0	0	0
	0	0	0	0	0	0	0	0	0	98	0	0	0	0	0	0	0	0
-12	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
-14	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
-16	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0

Fig. 8. Sample output matrix.



Fig. 9. Machine grid from simulated TIROS orbit.

MESOMETEOROLOGY PROJECT - - - - - RESEARCH PAPERS
(Continued from front cover)

16. Preliminary Result of Analysis of the Cumulonimbus Cloud of April 21, 1961 -Tetsuya Fujita and James Arnold

17. A Technique for Precise Analysis of Satellite Photographs - Tetsuya Fujita

18. Evaluation of Limb Darkening from TIROS III Radiation Data - S.H.H. Larsen, Tetsuya Fujita, and W. L. Fletcher

19. Synoptic Interpretation of TIROS III Measurements of Infrared Radiation -Finn Pedersen and Tetsuya Fujita

20. TIROS III Measurements of Terrestrial Radiation and Reflected and Scattered Solar Radiation - S.H.H. Larsen, Tetsuya Fujita, and W.L. Fletcher

21. On the Low-level Structure of a Squall Line - Henry A. Brown

22. Thunderstorms and the Low-level Jet - William D. Bonner

23. The Mesoanalysis of an Organized Convective System - Henry A. Brown

24. Preliminary Radar and Photogrammetric Study of the Illinois Tornadoes of April 17 and 22, 1963 - Joseph L. Goldman and Tetsuya Fujita

25. Use of TIROS Pictures for Studies of the Internal Structure of Tropical Storms -Tetsuya Fujita with Rectified Pictures from TIROS I Orbit 125, R/O 128 -Toshimitsu Ushijima

26. An Experiment in the Determination of Geostrophic and Isallobaric Winds from NSSP Pressure Data - William Bonner

27. Proposed Mechanism of Hook Echo Formation - Tetsuya Fujita with a Preliminary Mesosynoptic Analysis of Tornado Cyclone Case of May 26, 1963 -Tetsuya Fujita and Robbi Stuhmer

28. The Decaying Stage of Hurricane Anna of July 1961 as Portrayed by TIROS Cloud Photographs and Infrared Radiation from the Top of the Storm -Tetsuya Fujita and James Arnold

29. A Technique for Precise Analysis of Satellite Data, Volume II - Radiation Analysis, Section 6. Fixed-Position Scanning - Tetsuya Fujita

30. Evaluation of Errors in the Graphical Rectification of Satellite Photographs -Tetsuya Fujita

(Continued on outside)